Dynamic responses of rigs subject to drag forces: exact solution with discontinuous property and loading

Alexander N. Papusha[1], Tore M. Jonassen[2], Ove T. Gudmestad[3,4,5]

1. Murmansk State Technical University, Murmansk, Russia
2. Oslo University College, Oslo, Norway
3. Statoil, Stavanger, Norway
4. NTNU, Trondheim, Norway
5. University of Stavanger, Stavanger, Norway

Alexander.Papusha@mstu.edu.ru, torejo@hio.no, otg@statoil.com

Keywords: Offshore structure, drag forces, waves and currents, linearization, exact solution for drag loading on rigs, dynamics and ringing response.

Abstract: The authors developed symbolic solutions for responses of a single degree−of−freedom vibrating model depicting offshore structures subject to drag loading due to waves and currents. Currently, considerable interest in the oil extraction industry motivated only numerical (not analytical) solutions from Fourier expansion of loading and time integration of the governing equation.
***Mathematica* has empowered users to construct the exact nonlinear solution that we document here. The advantage is to capture terms, which are numerically intractable, in the case of discontinuities by selective appropriate value of integration constants of the homogenous solution.**
The authors present several symbolic and numeric solution details.

■ Introduction

The problem of interaction between offshore rigs and currents of flowing streams in water has been widely studied during last 30 years by well−known specialists on offshore technology. The offshore structure in question is a complicated aggregation. As a first approximation, for the purpose of design−analysis the motion of a rig is conceived as that of a simple harmonic oscillator subjected to drag forces generated by wave and current loading. Such r*esponses* have been extensively studied, both theoretically and numerically over the last several years.

Linearization of the *drag forcing term* was discussed by Gudmestad and Connor (1983) and the effects of considering different deterministic wave theories and integrating the total loading up to the *free surface* were discussed by Gudmestad and Poumbouras

(1988). An attempt to extend the analysis of the effects of *wave kinematics* to irregular seas was suggested by Gudmestad (1990).

Later, a number of procedures have been suggested to enhance the understanding of wave actions on slender offshore structures, see for example Skjelbreia et al., (1991), Stansberg and Gudmestad (1996) and Grue et al. (2003) regarding wave kinematics.

Ringing response aspects have been extensively studied in view of the so called "*ringing response*" of structures in the sea, Stansberg (1992), Jeffreys and Rainey (1994), Newman et al. (1996) and Krogstad et al. (1998). This ringing response is a transient response that seems to be caused by an impulse type of loading (see for example Kvitrud (1994) on structures subjected to either drag forcing terms associated with nonlinear kinematics, free surface effects or quadratic forcing terms (Lighthill, 1979).

It has been suggested that ringing mainly occurs in steep waves, see for example Chaplin et al. (1997) and Welch et al. (1999). These waves would give a type of loading resembling an impact. Reference is also made to Gurley and Kareem (1998), from where Figure 1a is copied. Figures 1b to 1d show a "ringing like response" of the Kvitebjoern jacket installed in the North Sea at 190 m water depth. This is a slender offshore jacket responding dynamically to wave loading.



Time history of ringing like response of the Kvitebjoern jacket installed in the North Sea at 190 m water depth. The figure shows from top to bottom the North−South accelerations (mm/s^2), velocities (mm/s) and displacements (mm) as function of time (s) for an event recorded on 01.01.2004 at 2 p.m.



It is, in general, agreed that the ringing phenomenon is not well understood,
the standards, for example Norsok Standard N−003 (1999), call for tank testing whenever ringing may be a feature in the design of spatially large offshore structures.

For structures dominated by drag type loading, careful dynamic analysis in the time domain is required.

Recent analysis of nonlinear drag loading type wave effects is presented by e.g. Quek et al. (2005).

■ Dynamic Models of an Offshore Slender Structure

The general second order ordinary differential equation for the horizontal response $y = y(t)$ of a single degree−of−freedom slender offshore structure subjected to constant nonlinear *drag loading* (*generated by the velocity of the current*) is according to experiments (given by the term per unit length of the structure; see for example Sarpkaya and Issacsson (1981)) is presented below:

$$m\,\frac{d^2y}{dt^2} + c\,\frac{dy}{dt} + k\,y = \frac{1}{2}\,\rho\,C_d\,D_0\,U(x,\,t)\,\left|\,U(x,\,t)\,\right|$$

Here the following notations are employed:

• m is the mass of the structure;

• k is the linear stiffness of the structure;

• c is a linear damper reaction coefficient;

• D_0 is the diameter of the slender structure;

• ρ is the density of the fluid (water);

• C_d is the drag coefficient for the flow;

• $U(x,\,t)$ and $|\,U(x,\,t)\,|$ are the velocity and the absolute value of the velocity U of the constant flow (current) past the structure, x – is a current coordinate of running waves.

This drag type loading is in general attributed to the shedding of vortices in the downstream flow direction of the current.

In the case of a combined wave and current loading, the nonlinear drag force according to Morison's postulate (Morison et al., 1950) for slender structures ($D_0/L \leq 0.2$, where L is the wave length) becomes:

$$m\,\frac{d^2y}{dt^2} + c\,\frac{dy}{dt} + k\,y = \frac{1}{2}\,\rho\,C_d\,D_0\,(U_0\,(x,\,t) + u(t))\,\left|\,U_0\,(x,\,t) + u(t)\,\right|$$

Here,

• $u(t)$ is the velocity of the oscillating flow;

• C_d is the modified drag coefficient for the combined flow. The magnitude of C_d exhibits a significant variation with Reynolds number (Re), Keulegan−Carpenter number (K) and relative roughness (k/D). In this paper we will treat C_d as a constant.

For the selection of values for C_d , in accordance with international recommendations, see e.g. Gudmestad and Moe (1996).

It should be noted, that in this analysis, the influence of the displacement of the structure on the flow is not accounted for see for example Gudmestad and Connor (1983) for a discussion of the "relative velocity effects".

In this paper we omit the mass forcing term while considering slender structures where the drag term is dominating, see for example Sarpkaya and Issacson (1981) criteria.

It should be noted that the mass term is a linearly proportional to $\frac{du(t)}{dt}$ and that would cause traditional dynamic amplification of the response at the resonance frequency, i.e. when $\omega = \Omega$. Further resonances will be triggered while accounting for the force contributions on the free surface of the wave(s).

Additional nonlinear loading terms as suggested, see for example Newman et al. (1996), will not be discussed here.

In the case of a sinusoidal (*wave*) loading caused by an oscillating force with frequency Ω, equation (2) for the case of drag loading reduces to:

$$m \frac{d^2 y}{dt^2} + c \frac{dy}{dt} + k y = A_1{}^2 \; Sin(\Omega t) \left| Sin(\Omega t) \right|$$

The value of the term A_1 is, for example, for the wave $\eta(x, t) = \eta_0 \; \mathrm{Sin}(k_1 \; x - \omega t)$ given as:

$$A_1{}^2 = \frac{1}{2} \rho C_d \; D_0 \left(\eta_0 \; \Omega \; \frac{\mathrm{Cosh}(k_0 \; (x + h))}{\mathrm{Sin}(k_0 \; h)} \right)$$

Here, k_0 is the wave number and η_0 is the surface wave amplitude according to linear wave theory.

For the case with no damping, this equation becomes:

$$\frac{d^2 y}{dt^2} + \omega^2 \; y = A^2 \; Sin(\Omega t) \; | \; Sin(\Omega t) \; |$$

The solution of equation (5) will be considered in detail in the forthcoming.

The solution of this equation will thereafter be extended to the solution of the damped equation (3) and subsequently to the solution of the case where the forcing function is represented by a combination of a wave and current (2) as given in (6):

$$\frac{d^2 y}{dt^2} + \omega^2 \; y = (A_0 \; + A_1 \; Sin(\Omega t)) \; | \; A_0 \; + A_1 \; Sin(\Omega t) \; |$$

As the wave height and the drag coefficient both are modified by the current, the term A_1 in equation (6) is different from A_1 as given by equation (4) above for the wave only case.

The case where the forcing function is represented by a combination of two sinusoidal terms (representing two waves acting simultaneously having different amplitudes and frequencies) as in (7) will finally be considered:

$$\frac{d^2 y}{dt^2} + \omega^2 \; y = (A_0 \; + A_1 \; Sin(\Omega_1 \; t) + A_2 \; Sin(\Omega_2 \; t)) \; | \; A_0 \; + A_1 \; Sin(\Omega_1 \; t) + A_2 \; Sin(\Omega_2 \; t) \; |$$

One could also continue to seek the solution in case of realistic stochastic seas with a large number of waves with different amplitudes A_i, frequencies Ω_i and phase shifts β_i acting on the structure.

■ Dynamic Response of a Rig: Traditional Approach

It's obvious case (see graphics below) that time function generated by a current load in a right side of differential equation (6) has a second time derivative discontinues property.

```
$TextStyle = {FontFamily → "Times−Italic", FontSize → 12};
SetOptions[Plot, DisplayFunction → Identity];
SetOptions[ParametricPlot, DisplayFunction → Identity];
SetOptions[Simplify, TimeConstraint → Infinity];

f[t_] = (A0 + A1 Sin[Ω t]) Sqrt[(A0 + A1 Sin[Ω t])^2];
g[t_] = D[f[t], t]; h[t_] = D[f[t], t, t];

popts = {A0 → .5, A1 → 1, Ω → .5};
grLoads =
    Plot[f[t] /. popts, {t, −14, 8}, AxesLabel → {"t", "y(t)"}, Ticks → None];
grFirstDerivative = Plot[g[t] /. popts, {t, −14, 8},
      AxesLabel → {"t", "y'(t)"}, Ticks → None];
grSecondDerivative = Plot[h[t] /. popts, {t, −14, 8},
      AxesLabel → {"t", "y''(t)"}, Ticks → None];

Show[GraphicsArray[{grLoads, grFirstDerivative, grSecondDerivative}]];
```



In order to find a continuos solution of the ODE with a right side function displaying discontinues properties one would match expand this function in to, for example, Fourier series (save as usual several main terms!) and solve this problem by well−known procedures: in symbolic or in numerical types. Then, one just convert the original singular problem into non singular problem having smooth solution and expressed that the smooth solution obtained by this usual simple way is a single and correct solution of the initial problem in question.

What are the results that will be obtained after this traditional manipulations?

Answers of this question we will get later after comparison of two solutions. The first type of solutions are corresponding to a traditional approach obtained by numerical procedure, and the second one are referring to symbolic solutions derived by *Mathematica* from the initial problem having a second derivative jagged curve (see graphics above).

First of all, let us consider the simplest form of the equation for the response caused by drag loading, equation (5):

$$\frac{d^2 y}{d t^2} + \omega^2 \; y = A^2 \; Sin(\Omega t) \mid Sin(\Omega t) \mid$$

The drag forcing term can be expanded in a Fourier series with a limited terms, see for example Gudmestad and Connor (1983) as:

```
dragForce[t_, M_] := Sum[2 / (π j) (1 − j^2 / (j^2 − 4)) Sin[j Ω t], {j, 1, M, 2}];
```

In example of the expansion of a load in Fourier series with saving a limited number of terms is presented below

dragForce[*t*, 7]

$$\frac{8\sin(t\,\Omega)}{3\,\pi} - \frac{8\sin(3\,t\,\Omega)}{15\,\pi} - \frac{8\sin(5\,t\,\Omega)}{105\,\pi} - \frac{8\sin(7\,t\,\Omega)}{315\,\pi}$$

It should be noted that only the odd harmonic functions are included in (8) as the term is symmetric (mirrored) around origo. In case we include a constant velocity component given by U_0, the even harmonics will contribute as well. When we find the total force on the structure, we have to integrate force contributions to the free surface of the waves. Odd harmonic components emerge during the integration to the free surface; see for example Gudmestad and Connor (1983).

We have not carried out any evaluation of the convergence of the Fourier expansion of the loading term to find whether it converges towards the exact drag loading term. It should in particular be noted that the derivative and in particular the second derivatives of the Fourier expansion of the loading term given by **dragForce[t,7]** are different from the derivatives of the exact drag force.

It is known (see for example Carleson, 1966) that if a function is L^2 (quadratic integrable), then the Fourier series of the function converges "almost everywhere", that is everywhere, except possibly on a set of Lebesgue measure zero. Furthermore, the Fourier expansion of a function may converge towards the function, while the derivative of the Fourier expansion may not necessarily converge towards the derivative of the function itself. Even if the Fourier series of the loading term converges towards the exact loading term, the resulting solution of the equation when using the Fourier expansion of the loading term may thus not converge towards the exact solution of the original equation.

□ The Approximate Solution of General Equation

Part one: Direct numerical time history method of evaluation

The solution of a differential equation (8) can be found numerically by time history integration, using, for example, a Runge–Kutta integration scheme. Using the integration scheme inherent in *Mathematica* Solver (Wolfram, 2003), we have calculated the time history development of the response for equation (8) . The initial conditions and the values of the constants are chosen arbitrarily, we will later show the effect at resonances, $\omega = \Omega$. The results including the phase diagram for the solution are shown in graphics.

```
eqInitial[s_] := s''[t] + 2 h s'[t] + ω^2 s == A^2 Sin[Ω t] Sqrt[Sin[Ω t]^2];
s = y[t]; solNumerical =
  NDSolve[{eqInitial[s] /. {h → 0, A → 1, ω → 1.1, Ω → .5}, y(0) == 1, y'(0) == 0},
    s, {t, 0, 130}] // Flatten;

grNumerical1 =
  Plot[Evaluate[(y(t) /. solNumerical)], {t, 0, 30}, AxesLabel → {"t", "y(t)"}];
w2 = D[y(t) /. solNumerical, t]; gr6 = ParametricPlot[
    Evaluate[{y(t) /. solNumerical, w2}], {t, 0, 130}, AxesLabel → {"y(t)", "y'(t)"}];
Show[GraphicsArray[{grNumerical1, gr6}]];
```



It should be noted that the resulting solution as function of time should be analyses further to check whether other solution schemes, possibly with finer integration steps, would reveal different solutions to the problem.

Part two: Approximate solution with a Fourier time function expanding

Equation (5) with the loading function expanded in a Fourier series as given above, reveals a symbolic solution that is quite similar to the solution found above by numerical integration of the exact equation.

```
sol11 =
  DSolve[{(y''[t] + 2 h y'[t] + ω^2 y[t] == dragForce[t, 7]) /. {h → 0, s → y[t]}, y[0] == 1,
    y'[0] == 0}, y[t], t] // Flatten // Simplify
```

$$\{y(t) \to (315\,\pi\,\omega\,(\omega^8 - 84\,\Omega^2\,\omega^6 + 1974\,\Omega^4\,\omega^4 - 12916\,\Omega^6\,\omega^2 + 11025\,\Omega^8)\cos(t\,\omega) - 8\,(20\,\Omega\,(\omega^6 - 143\,\Omega^2\,\omega^4 + 5371\,\Omega^4\,\omega^2 - 53613\,\Omega^6)\sin(t\,\omega) + 2\,\omega\,(\cos(6\,t\,\Omega)\,\omega^6 - 40\,\omega^6 + 3464\,\Omega^2\,\omega^4 - 35\,\Omega^2\cos(6\,t\,\Omega)\,\omega^4 - 84760\,\Omega^4\,\omega^2 + 259\,\Omega^4\cos(6\,t\,\Omega)\,\omega^2 + 565176\,\Omega^6 + (25\,\omega^6 - 1787\,\Omega^2\,\omega^4 + 29035\,\Omega^4\,\omega^2 - 27273\,\Omega^6)\cos(2\,t\,\Omega) + 4\,(\omega^6 - 53\,\Omega^2\,\omega^4 + 439\,\Omega^4\,\omega^2 - 387\,\Omega^6)\cos(4\,t\,\Omega) - 225\,\Omega^6\cos(6\,t\,\Omega))\sin(t\,\Omega))) / (315\,\pi\,\omega\,(\omega^8 - 84\,\Omega^2\,\omega^6 + 1974\,\Omega^4\,\omega^4 - 12916\,\Omega^6\,\omega^2 + 11025\,\Omega^8))\}$$

Unlike the numerical procedure presented in the previous subsection resonance conditions are derived below in symbolic form.

```
Part[Solve[(ω^8 - 84 Ω^2 ω^6 + 1974 Ω^4 ω^4 - 12916 Ω^6 ω^2 + 11025 Ω^8) == 0, Ω],
  {5, 6, 7, 8}] // Flatten
```

$$\left\{\Omega \to \frac{\omega}{7}, \Omega \to \frac{\omega}{5}, \Omega \to \frac{\omega}{3}, \Omega \to \omega\right\}$$

The graphics below identify to a similar response of the structure as obtained by the previous numerical procedure.

```
popts = {ω → 1.1, Ω → .5};
grNumerical11 =
  Plot[Evaluate[(y(t) /. sol11) /. popts], {t, 0, 30}, AxesLabel → {"t", "y(t)"}];
v2 = D[y(t) /. sol11, t]; grPhase = ParametricPlot[
    Evaluate[{y(t) /. sol11, v2} /. popts], {t, 0, 130}, AxesLabel → {"y(t)", "y'(t)"}];
Show[GraphicsArray[{grNumerical11, grPhase}]];
```



Just the first four terms of the Fourier expanded loading function have been taken in to consideration and a fine approximation of time response have been obtained.

It should be noted that for the case of the Fourier expanded loading term, we could obtain an analytical solution to the required level of accuracy by using the Fourier expanded loading term to the same number of terms.

What remains uncertain, however, is whether the solution converges to the solution of the exact equation, bearing in mind that the Fourier expansion is an approximation to the nonlinear loading function?

■ Dynamic Response of a Rig: *Mathematica* Evaluation

□ Exact solution of the non damped equation for drag wave loading on slender offshore structure

We have attempted to find an exact analytical closed form solution for equation (5). As far as we are concerned, such a solution has not previously been presented in the literature in this context, only numerical or Fourier methods has been used.

As the numerical time series solution in practice gives exactly the same solution as for the equation with the loading term expanded in a Fourier series in accordance with equation (13), it has been the general impression that the solution to the problem has been identified. This has lead to the suggestion that this solution to equation (5) exhibits resonance terms for the odd harmonics, i.e. there will be resonances in the case where n is an odd number.

We have challenged this suggestion and have tried to identify an analytical solution to (5) by applying the equation solver of the computer code *Mathematica* (Wolfram, 2003). It proved necessary, however, to rewrite equation (5) in order to identify the solution. The *Mathematica* Equation Solver would not work with absolute values of the forcing term that had to be rewritten into the form:

$$\text{Sin}[\Omega\, t]\ \text{Abs}[\text{Sin}[\Omega\, t]\,] \equiv\ \text{Sin}[\Omega\, t]\,\sqrt{\text{Sin}[\Omega\, t]^{\,2}}$$

Having formulated the problem: To find an analytical closed form solution to equation (5) with the loading term reformulated as given in (9), the *Mathematica* differential equation solver **DSolve** identified the analytical solution presented in equation (5) below:

```
solExact = DSolve[{eqInitial[s] /. h → 0, y(0) == 1, y'(0) == 0}, s, t] // Flatten // Simplify
```

$$\left\{y(t) \to \frac{1}{2\,\omega^2\,(\omega^2\,-4\,\Omega^2)}\left(\left(-\cos(2\,t\,\Omega)\,\omega^2\,+\omega^2\,-4\,\Omega^2\right)\csc(t\,\Omega)\,\sqrt{\sin^2(t\,\Omega)}\;A^2\,+\right.\right.$$
$$\left.\left.2\left(\omega^4\,-4\,\Omega^2\;\omega^2\,+2\,A^2\;\Omega\,\sqrt{\Omega^2}\,\right)\cos(t\,\omega)\right)\right\}$$

One have to emphasize a several important points from the symbolic solution:

- Only one resonance condition follows from the evaluation (Not a set of resonances!).
- Discontinuity property of motion is obtained from the solution obtained by *Mathematica*, at the points $t_i = i\,\pi/\Omega$, $i \in \mathbb{N}$. (See comments below).

Unlike the previous symbolic solution (see the Fourier expansion) the *only* resonance condition for the original problem can be found

```
Part[Solve[ (ω^2 − 4 Ω^2 ) == 0, Ω], 2]
```

$$\left\{\Omega \to \frac{\omega}{2}\right\}$$

The symbolic solution derived by DSolve above exhibits a *discontinuity at the points where* $\sin(\Omega t) = 0$. From general theory for ODE's this can not be correct as long as the driving force is C^1, and we conclude that the symbolic solution found by *Mathematica* using a simple **DSolve** command is valid only in the interval $[0, \pi/\Omega)$. We therefore have to investigate the discontinuity of the second derivative of the driving force at $Sin(\Omega t) = 0$ further.

The visualization of the solution found by *Mathematica* by a simple **DSolve** command as an example from the symbolic evaluation is presented in the graphics below, with $A = 1,\ \omega = 1.1,\ \Omega = .5$.

```
popts = {A → 1, ω → 1.1, Ω → .5};
grExact =
    Plot[Evaluate[(y(t) /. solExact) /. popts], {t, 0, 30}, AxesLabel → {"t", "y(t)"}];
u2 = D[y(t) /. solExact, t]; grPhaseExact =
  ParametricPlot[Evaluate[{y(t) /. solExact, u2 } /. popts],
    {t, 0, 130}, AxesLabel → {"y(t)", "y'(t)"}];
Show[GraphicsArray[{grExact, grPhaseExact}],
    DisplayFunction → $DisplayFunction];
```



In spite of a discontinues property of the symbolic solution obtained by *Mathematica* it's possible to use the symbolic solution obtained by *Mathematica* and obtain the exact solution by a "gluing procedure" presented below.

Let us first give an analytical expression for the to equation (5) with the forcing term $f(t) = A^2 \sin(\Omega t)\,|\sin(\Omega t)|$. The function $f''(t)$ is discontinuous at the points $t_i = i\pi/\Omega$ for $i \in \mathbb{N}$. Hence we write $\mathbb{R}^+$ as a disjoint union of half−open intervals $I_i = [(i-1)\pi/\Omega,\ i\,\pi/\Omega)$ as

$$\mathbb{R}^+ = \bigcup_{i \geq 1} I_i$$

By applying the variation of parameters method and by gluing together solutions of the corresponding homogeneous equation in the endpoints if the intervals I_i it is not so hard to show the following result by pen and paper, the analytical expression for the solution of (5):

Given the initial value problem

$$\frac{d^2 y}{dt^2} + \omega^2\, y = A^2\, sin(\Omega t)\,|\,sin(\Omega t)\,| \ \text{ where } y(0) = \alpha \ \text{ and } y'(0) = \beta$$

for $t \in \mathbb{R}^+$ and $\omega > 0$ and $\Omega > 0$. Let $I_j = [(j-1)\pi/\Omega,\ j\pi/\Omega)$ where $j \in \mathbb{N}$. Then the unique solution of the initial problem in the half−open interval I_j is given as

$$y(t) = s_j(t) = p_j(t) + k \sum_{i=1}^{j-1} (-1)^i\, \tau_i(t) \ \text{ for } j \geq 2$$

and as

$$y(t) = s_1(t) = p_1(t) \text{ on } I_1$$

provided $\omega^2 \neq 4\,\Omega^2$. Here the functions $p_j(t)$ and $\tau_j(t)$ and the constant k are given by

$$p_j(t) = \frac{(-1)^{j+1}\, A^2\,(\omega^2 - 4\,\Omega^2 - \omega^2 \cos(2\,\Omega\,t))}{2\,\omega^2\,(\omega^2 - 4\,\Omega^2)} + \frac{(2\,\Omega^2\, A^2 + \alpha\,\omega^2\,(\omega^2 - 4\,\Omega^2))\cos(\omega\,t) + \beta\,\omega(\omega^2 - 4\,\Omega^2)\sin(\omega\,t)}{\omega^2\,(\omega^2 - 4\,\Omega^2)},$$

$$\tau_j(t) = \cos\left(\omega\left(t - \frac{j\,\pi}{\Omega}\right)\right)$$

and

$$k = \frac{4\,A^2\,\Omega^2}{\omega^2\,(\omega^2 - 4\,\Omega^2)}.$$

Let us now compare the exact solution above with the numerical solution obtained by **NDSolve**. Clearly the solution above may be written in *Mathematica* as:

```
p[j_, A_, Ω_, ω_, α_, β_][t_] :=
  (-1)^(j+1) (ω^2 - 4 Ω^2 - ω^2 Cos[2 Ω t]) A^2 / (2 ω^2 (ω^2 - 4 Ω^2)) +
    1 / (ω^2 (ω^2 - 4 Ω^2)) ((2 Ω^2 A^2 + α ω^2 (ω^2 - 4 Ω^2)) Cos[ω t] + β ω (ω^2 - 4 Ω^2) Sin[ω t]);
k[A_, Ω_, ω_] := 4 A^2 Ω^2 / (ω^2 (ω^2 - 4 Ω^2));
τ[j_, Ω_, ω_][t_] := Cos[ω (t - j π / Ω)];
ss[1, A_, Ω_, ω_, α_, β_][t_] := p[1, A, Ω, ω, α, β][t];
ss[j_, A_, Ω_, ω_, α_, β_][t_] :=
  p[j, A, Ω, ω, α, β][t] + k[A, Ω, ω] Sum[(-1)^i τ[i, Ω, ω][t], {i, 1, j-1}];
```

We will use the parameter values $A = 1$, $\omega = 0.55$ and $\Omega = 0.5$, the initial conditions $y(0) = \alpha = 1$ and $y'(0) = \beta = 0$, and compare the solutions on the interval $t \in [0,\ 100\,\pi]$. The numerical solution is found in the usual way:

```
numsol =
  NDSolve[{y''[t] + (0.55)^2 y[t] == Sin[0.5 t] Abs[Sin[0.5 t]], y[0] == 1, y'[0] == 0},
    y[t], {t, 0, 100 π}] // Flatten;
```

Here is the graph of the exact solution in the interval $[0,\ 100\,\pi]$:

```
pl = {};
Do[AppendTo[pl, Plot[ss[i, 1, 0.5, 0.55, 1, 0][t],
    {t, 2 (i - 1) π, 2 i π}]], {i, 1, 50}];
Show[pl, DisplayFunction → $DisplayFunction,
  AxesLabel → {"t", "y(t)"}];
```



A graphical comparison between the numerical solution and the exact expression for the solution can be given as follows:

```
pl = {};
Do[AppendTo[pl, Plot[(y[t] /. numsol) – ss[i, 1, 0.5, 0.55, 1, 0][t],
        {t, 2 (i – 1) π, 2 i π}, PlotStyle → RGBColor[1, 0, 0]]], {i, 1, 50}];
Show[pl, DisplayFunction → $DisplayFunction, AxesLabel → {"t", "Δy"}];
```



As seen above, the maximal difference between the numerical solution and the exact solution is of order 10^{-6}. In a section below we show how to compute the exact solution with *Mathematica* using a limit and gluing construction.

The following shows how to construct an exact solution for a class of second order equations purely with *Mathematica*. The following leads us to a solution by *Mathematica*.

We consider a linear second order ODE of the form $L(y) = f$ where f'' has a discrete set of discontinuities $\{x_1, x_2, x_3, \ldots, x_n\} \subset I \subset \mathbb{R}$ where I is some closed interval on the real line. We assume that the solution of the equation is expressible in a closed form on each subinterval (x_i, x_{i+1}). In this case it can be shown that the solution is at most C^3 on

I, and hence the expression for the exact solution must be given as piecewise expressions of functions in a closed form. Let $I = [x_0, x_{n+1}]$ and consider the following partition of I: $x_0 < x_1 < x_2 \cdots < x_n < x_{n+1}$. By applying *Mathematica* to the problem $L(y) = f$, $y(x_0) = \alpha$, $y'(x_0) = \beta$ we have observed that **DSolve** only give the correct solution in the subinterval $[x_0, x_1)$, and hence fails to give the correct solution on the whole interval I. One posibillity is to compute the one−sided limits of $y = y(t)$ and $y' = y'(t)$ in the right endpoints if the subintervals $[x_i, x_{i+1})$, starting from the leftmost subinterval, and apply **DSolve** consequtively. However, such a method is clearly non−elegant, and an over−kill of the problem, as the difference between *Mathematica's* solution and the exact solution is a solution of the homogenous problem $L(y) = 0$. Hence the problem should reduce to an algebraic problem at the points $\{x_i\}$ inside the interval I. The following code in *Mathematica* that will handle such kind of problems. It computes four one−sided limits at each point x_i and then apply **Solve** to adjust the exact solution. The function returns a **Piecewise** function object, and hence it is possible to treat this object like any other function defined in *Mathematica*. The code runs slow. We have tested the code on a DELL laptop, 1.13 GHz CPU with 256 Mb RAM and on a P4 workstation 3.0 GHz CPU, 4Gb RAM, and of course it runs much faster on the workstation. We have also tested for various lengths of the "singularity list". We have tried to remove the **Simplify** commands inside the code, but this slows down the performance for longer "singularity lists", but increases the performance for short "singularity lists".

A short explanation of the function **GenerateSolution** below.The variable **psol** is the particular solution of the problem in the interval $[x_0, x_1)$ comming from **DSolve**. The variable **ghomsol** is the general solution of the homogenuous problem also computed by **DSolve**. **var** is the symbol for the free variable in the solution. **singularities** is a list of the discontinuties including the initial point and the endpoint in the interval I. **asmt** is a set of assumtions that is passed to **Limit**. We need to pass assumtions to **Limit** if the equation contains constants in symbolic form in order to get *Mathematica* to compute correctly, or else *Mathematica* assumes that the constants are complex numbers, resulting in unwanted output.

```
GenerateSolutions[psol_, ghomsol_,
   var_, singularities_List, asmt_List] :=
  Module[{funclist = {psol}, derhoneq, derfunc, lefunc,
    rifunc, leftderfunc, rightderfunc, newconsts,
    addsol, j = Length[singularities] - 2},
   derhoneq = D[ghomsol, var];
   Do[
    (derfunc = Simplify[Evaluate[D[funclist[[i]], var]]];
     lefunc = Limit[funclist[[i]], var → singularities[[i + 1]],
       Assumptions → asmt, Direction → 1];
     rifunc = Limit[funclist[[i]], var → singularities[[i + 1]],
       Assumptions → asmt, Direction → -1];
     leftderfunc = Limit[derfunc, var → singularities[[i + 1]],
       Assumptions → asmt, Direction → 1];
     rightderfunc = Limit[derfunc, var → singularities[[i + 1]],
       Assumptions → asmt, Direction → -1];
     newconsts = Flatten[Solve[{Evaluate[ghomsol ==
              lefunc - rifunc /. var → singularities[[i + 1]]],
           Evaluate[derhoneq == leftderfunc - rightderfunc /. var →
              singularities[[i + 1]]]}, {C[1], C[2]}] // Simplify];
     addsol = Simplify[ghomsol /. newconsts];
     AppendTo[funclist, funclist[[i]] + addsol];), {i, 1, j}];
   Return[
    Piecewise[Table[{funclist[[i]], singularities[[i]] ≤ var &&
        var < singularities[[i + 1]]}, {i, 1, j + 1}]]]
  ];
```

We will apply this construction to our sample equation above. First we will need the expression for a particular solution, and the expression for the general homogenous solution:

```
particular = y[t] /. Simplify[
    Flatten[DSolve[{y''[t] + ω^2 y[t] == A^2 Sin[Ω t] Sqrt[Sin[Ω t]^2],
       y[0] == α, y'[0] == β}, y[t], t]]];
homogene = v[t] /. Flatten[DSolve[v''[t] + ω^2 v[t] == 0, v[t], t]];
```

We will solve equation (5) in the interval $[0, 9\pi/\Omega)$. The output is supressed as the **Piecewise** function output is hard to format:

```
ff[t_] = GenerateSolutions[
   particular, homogene, t, Table[i π / Ω, {i, 0, 9}],
   {Ω > 0, ω > 0, A > 0, α ∈ Reals, β ∈ Reals}];
```

The exact solution at the interval $t\in[0, 9\pi/\Omega]$ is presented graphically below.

We will use the same parameters used in the previous evaluations.

```
popts = {A → 1, ω → 1.1, Ω → 0.5, α → 1, β → 0};
```

The following plot based on the exact solution above obtained by *Mathematica*:

```
pwexact = Plot[ff[t] /. popts, {t, 0, 18 π}, AxesLabel → {"t", "y(t)"}];
Show[pwexact, DisplayFunction → $DisplayFunction];
```



Numerical solution is repeated for feather :

```
grNumerical2 =
    Plot[Evaluate[(y[t] /. solNumerical)], {t, 0, 18 π}, AxesLabel → {"t", "y(t)"}];
```

A plot ofthe exact symbolic solution together with the numerical solution are plotted together below.

```
Show[pwexact, grNumerical2, DisplayFunction → $DisplayFunction];
```



As seen above, there is no significant difference between the symbolic and the numerical solution.

The numerical solution in the phasespace is presented below.

```
grNumerical3 =
  Plot[Evaluate[(y(t) /. solNumerical)], {t, 0, 18 π}, AxesLabel → {"t", "y(t)"}];
w2 = D[y(t) /. solNumerical, t]; gr6 = ParametricPlot[
    Evaluate[{y(t) /. solNumerical, w2 }], {t, 0, 18 π}, AxesLabel → {"y(t)", "y'(t)"}];
Show[GraphicsArray[{grNumerical3, gr6}], DisplayFunction → $DisplayFunction];
```



The corresponding exact solution in the phasespace presented graphicly.

```
b1 = ParametricPlot[Evaluate[{ff[t], ff'[t]} /. popts],
      {t, 0, 18 π}, AxesLabel → {"y(t)", "y'(t)"}];
Show[b1, DisplayFunction → $DisplayFunction];
```



As in the previous evaluations no differences are observed in the phaseapace diagrams either of course.

Finally, we will investigate the *resonance conditions for the exact solution*.

As seen from symbolic evaluation, there is no resonance at the higher *odd harmonics* as found when using a Fourier expansion of the loading term. The resonance condition will now be for, that is for the case $\Omega \to \frac{\omega}{2}$, this is at the second harmonics of the natural frequency of the oscillations of the structure, when the period for the oscillating loading is twice the natural period for the structural oscillations ($T_{\text{Loading}} = 2\, T_{\text{structure}}$). This is in confirmation of the results of some investigators; see for example Kvitrud (1994); who have pointed to a second harmonics resonance in the case of the ringing phenomenon.

□ Exact solution of the general damper model

The new equation in the *case of damping* derived from (2) is presented below

$$\frac{d^2 y}{dt^2} + 2\,h\,\frac{dy}{dt} + \omega^2\, y = A^2\, Sin(\Omega\, t)\,\sqrt{Sin(\Omega\, t)^2}$$

as well one finds the exact solution to the modified equation (10) by the following

solExactDamper = DSolve[{eqInitial[*s*], *y*(0) == 1, *y*′ (0) == 0}, *s*, *t*] // Flatten // Simplify

$$\{y(t) \to \left(e^{-t\left(h+\sqrt{h^2-\omega^2}\right)}\right.$$
$$\left(e^{2t\sqrt{h^2-\omega^2}}\ h\,\omega^6 - h\,\omega^6 + e^{2t\sqrt{h^2-\omega^2}}\ \sqrt{h^2-\omega^2}\ \omega^6 + \sqrt{h^2-\omega^2}\ \omega^6 -\right.$$
$$8\,e^{2t\sqrt{h^2-\omega^2}}\ h\,\Omega^2\,\omega^4 + 8\,h\,\Omega^2\,\omega^4 - 8\,e^{2t\sqrt{h^2-\omega^2}}\ \sqrt{h^2-\omega^2}\ \Omega^2\,\omega^4 -$$
$$8\sqrt{h^2-\omega^2}\ \Omega^2\,\omega^4 + 16\,e^{2t\sqrt{h^2-\omega^2}}\ h\,\Omega^4\,\omega^2 -$$
$$16\,h\,\Omega^4\,\omega^2 + 16\,e^{2t\sqrt{h^2-\omega^2}}\ \sqrt{h^2-\omega^2}\ \Omega^4\,\omega^2 +$$
$$16\sqrt{h^2-\omega^2}\ \Omega^4\,\omega^2 + 16\,e^{2t\sqrt{h^2-\omega^2}}\ h^3\,\Omega^2\,\omega^2 - 16\,h^3\,\Omega^2\,\omega^2 +$$
$$16\,e^{2t\sqrt{h^2-\omega^2}}\ h^2\,\sqrt{h^2-\omega^2}\ \Omega^2\,\omega^2 + 16\,h^2\,\sqrt{h^2-\omega^2}\ \Omega^2\,\omega^2 +$$
$$6\,A^2\,e^{2t\sqrt{h^2-\omega^2}}\ h\,\Omega\,\sqrt{\Omega^2}\ \omega^2 - 6\,A^2\,h\,\Omega\,\sqrt{\Omega^2}\ \omega^2 +$$
$$2\,A^2\,e^{2t\sqrt{h^2-\omega^2}}\ \sqrt{h^2-\omega^2}\ \Omega\,\sqrt{\Omega^2}\ \omega^2 + 2\,A^2\,\sqrt{h^2-\omega^2}\ \Omega\,\sqrt{\Omega^2}\ \omega^2 -$$
$$A^2\,e^{t\left(h+\sqrt{h^2-\omega^2}\right)}\ \sqrt{h^2-\omega^2}\ \cos(t\,\Omega)\,(8\,h\,\Omega + (\omega^2 - 4\,\Omega^2)\cot(t\,\Omega))$$
$$\sqrt{\sin^2(t\,\Omega)}\ \omega^2 + A^2\,e^{t\left(h+\sqrt{h^2-\omega^2}\right)}\ \sqrt{h^2-\omega^2}\ \csc(t\,\Omega)\sqrt{\sin^2(t\,\Omega)}$$
$$\left.\left(\omega^4 - 8\,\Omega^2\,\omega^2 + (\omega^2 - 4\,\Omega^2)\sin^2(t\,\Omega)\,\omega^2 + 16\,\Omega^2\,(h^2+\Omega^2)\right) -\right.$$
$$8\,A^2\,e^{2t\sqrt{h^2-\omega^2}}\ h\,\Omega^3\,\sqrt{\Omega^2} + 8\,A^2\,h\,\Omega^3\,\sqrt{\Omega^2} -$$
$$8\,A^2\,e^{2t\sqrt{h^2-\omega^2}}\ \sqrt{h^2-\omega^2}\ \Omega^3\,\sqrt{\Omega^2} - 8\,A^2\,\sqrt{h^2-\omega^2}\ \Omega^3\,\sqrt{\Omega^2} -$$
$$8\,A^2\,e^{2t\sqrt{h^2-\omega^2}}\ h^3\,\Omega\,\sqrt{\Omega^2} + 8\,A^2\,h^3\,\Omega\,\sqrt{\Omega^2} -$$
$$\left.\left.8\,A^2\,e^{2t\sqrt{h^2-\omega^2}}\ h^2\,\sqrt{h^2-\omega^2}\ \Omega\,\sqrt{\Omega^2} - 8\,A^2\,h^2\,\sqrt{h^2-\omega^2}\ \Omega\,\sqrt{\Omega^2}\right)\right)\Big/$$
$$\left.\left(2\,\omega^2\,\sqrt{h^2-\omega^2}\ (\omega^4 - 8\,\Omega^2\,\omega^2 + 16\,\Omega^2\,(h^2+\Omega^2))\right)\right\}$$

Resonance condition depending on h (coefficient of damping) is following from the symbolic solution of an algebraic equation presented below.

Solve[($\omega^4 - 8\,\Omega^2\,\omega^2 + 16\,\Omega^2\,(h^2+\Omega^2)$) == 0, Ω][[2, 1]]

$$\Omega \to \frac{1}{2}\sqrt{-2\,h^2 + \omega^2 - 2\sqrt{h^4 - h^2\,\omega^2}}$$

The following illustrate the exact solution of the damped equation with $h = 1$, $A = 1$, $\Omega = 1/2$ and $\omega = \sqrt{3}$ on the interval $[0,\ 20\,\pi)$ using the gluing construction with **`GenerateSolutions`** given above. We have here used the initial conditions $y(0) = 1$ and $y'(0) = 0$. As in our previous examples we use **`DSolve`** to find a particular solution, and to find the general homogenuous solution:

```
particular = (y[t] /. Flatten[
    DSolve[{y''[t] + 2 y'[t] + 3 y[t] == Sin[t / 2] Sqrt[Sin[t / 2]^2],
      y[0] == 1, y'[0] == 0}, y[t], t] // Simplify]);
homogene = (v[t] /. Flatten[DSolve[
     v''[t] + 2 v'[t] + 3 v[t] == 0, v[t], t]]);
```

The exact solution on $[0, 4\,\pi)$ represented as a **`Piecewise`** object is found as follows:

```
gg[t_] = GenerateSolutions[particular,
   homogene, t, {0, 2 π, 4 π}, {Automatic}];
```

The following displays a supressed form of the solution:

```
Short[gg[t] // Simplify, 4]
```

$$\begin{cases} \frac{1}{48} e^{-t} \left(46 \cos\left(\sqrt{2}\ t\right) - 3 e^{t} \cos\left(\frac{t}{2}\right) \sqrt{2 - 2\cos(t)} \left(\cot\left(\frac{t}{2}\right) + 2\right) + 4 e^{t} \sqrt{2 - 2\cos(t)} \csc\left(\frac{t}{2}\right) + 3 e^{t} \sqrt{2 - 2\cos(t)} \sin\left(\frac{t}{2}\right) + 26\sqrt{2} \sin\left(\sqrt{2}\ t\right)\right) & 0 \le t < 2\pi \\ \frac{1}{48} e^{-t} \left(\ll 1 \gg\right) & 2\pi \le t < 4\pi \end{cases}$$

The exact solution on $[0, 20\pi)$ represented as a **Piecewise** object is found as follows:

```
lgg[t_] = GenerateSolutions[particular,
   homogene, t, Table[2 i π, {i, 0, 10}], {Automatic}];
```

The graph of the exact solution on $[0, 20\pi]$ is given below:

```
Plot[lgg[t], {t, 0, 20 π}, AxesLabel → {"t", "y(t)"},
  DisplayFunction → $DisplayFunction];
```



A "jumping" motion of a damped model in a resonance condition $\Omega \approx \frac{\omega}{2}$ are presented at the graphics below, however this is due to the fact that the result returned by **DSolve** also in this case only is a solution of the initial value problem only in the interval $[0,\ 20\pi/11)$. Hence the plot shows a discontinuity at $t = 20\pi/11 \approx 5.71199$:

```
popts = {A → 1, ω → 1.1, Ω → .55, h → .51};
Off[ParametricPlot::"pptr"];
grExactDamper = Plot[Evaluate[(y(t) /. solExactDamper) /. popts],
      {t, 0, 3π}, AxesLabel → {"t", "y(t)"}];
w2 = D[y(t) /. solExactDamper, t]; grPhaseExactDamper =
  ParametricPlot[Evaluate[{y(t) /. solExactDamper, w2 } /. popts],
    {t, 0, 3π}, AxesLabel → {"y(t)", "y'(t)"}];
Show[GraphicsArray[{grExactDamper, grPhaseExactDamper}],
    DisplayFunction → $DisplayFunction];
```



□ Exact solution of a general non−damped model for multi−harmonic drag wave and current loading on slender offshore structure

We have, furthermore, established the closed form solution for the general non−damped equation under the combined wave and current loading on slender offshore structures, which means equation (7)

We will explore the situation where two waves are acting simultaneously as given by equation (7) later. The case of no damping $c = 0$ and when the wave frequencies are multiplies of each other such that $\Omega_1 = \Omega$ and $\Omega_2 = 2\,\Omega$ will be discussed in details.

The equation (7) has a closed form symbolic solution, the output is supressed as it covers almost two pages, note that it might also be necessary to increase the **TimeConstraint** option for **Simplify** on some machines depending on the CPU speed. It should be noted that the following computation could take a considerable amount of time, depending on the computertype and platform. The computation time will be dramatically reduced by removing the **Simplify** command at the end below :

```
solMultiHarmonic =
 DSolve[{(d^2 s)/dt^2 + ω^2 s == (A0 + A1 Sin(Ω1 t) + A2 Sin(Ω2 t)) Sqrt[
                                  (A0 + A1 Sin(Ω1 t) + A2 Sin(Ω2 t))^2 ]) /. {s → y[t]},
   y(0) == 1, y'(0) == 0}, y(t), t] // Flatten // Simplify;
```

A set of a new resonances conditions are derived from the algebraic equation and they are presented below

```
Part[Solve[(2 ω^2 A0 (A0 + sin(t Ω1) A1 + sin(t Ω2) A2) (ω^4 − 5 Ω1^2 ω^2 + 4 Ω1^4)
      (ω^4 − 5 Ω2^2 ω^2 + 4 Ω2^4) (Ω1^4 − 2 (ω^2 + Ω2^2) Ω1^2 + (ω^2 − Ω2^2)^2)) == 0, ω],
  {5, 6, 8, 11, 12, 13, 14}] // Flatten // Simplify
```

$$\{\omega \to \Omega_1,\ \omega \to 2\,\Omega_1,\ \omega \to \Omega_1 - \Omega_2,\ \omega \to \Omega_2,\ \omega \to 2\,\Omega_2,\ \omega \to \Omega_2 - \Omega_1,\ \omega \to \Omega_1 + \Omega_2\}$$

From last output follows that beside the subharmonic resonances $\Omega_i \approx \frac{\omega}{2}$, (one has met this resonances in the previous section), a main resonances $\Omega_i \approx \omega$ and combinations resonances likes $\Omega_1 \pm \Omega_2 \approx \omega$ occurs in the system.

When $A_2 = 0$, we have the situation with only one wave with frequency Ω. A resonance situation is obtained for $\Omega = 1/2\ \omega$.

- In case $A_1 = 0$, then we also have a situation with only one wave, now with frequency 2Ω and a resonance is obtained for $\omega = 2\,\Omega$. Equation (7) can be deducted by transferring the frequency 2Ω to Ω.
- For the case when A_1 and A_2 are not equal to 0, we obtain a mixed term with resonances at $\omega = 3\,\Omega$ and $\omega = \Omega$. The resonances are thus at $\omega = \Omega$ and at $\omega = 3\,\Omega$, that is at the sum and difference frequencies.
- In case we studied a more general loading function with two waves having frequencies Ω_1 and Ω_2, respectively, we would get resonances at sum and difference frequencies : $\Omega_1 + \Omega_2$ and at $\Omega_1 - \Omega_2$

- The solution to the two wave case where the wave frequencies are at a multiple of each other, Ω and 2Ω leads to resonances for the response at $\omega = \Omega$ and at higher multiples: $\omega = 2\,\Omega$, $\omega = 3\,\Omega$ and at $\omega = 4\,\Omega$.

■ Conclusions

Here we have initiated a new approach to the classical problem in offshore structural mechanics by introducing exact solutions of a classical model. Existing literature encompasees only Fourier series expansions and numerical approximations. This work also shows that a more sophisticated modeling than a simple one−dimensional one is in order to solve the "ringing" problem. Clearly, with the forthcomming development of oil and gas resources in the Barent Sea, exposed to hostile natural conditions, such studies are essential. We have also pointed out the importance of examining outputs from *Mathematica'*s standard differential equation solver, **`DSolve.`** We have furnished procedures for handling differential equations with higher order derivatives of forcing functions with discontinuities.

■ Reference

[1]. Carleson, L.: "On convergence and growth of partial sums of Fourier series", Acta Mathematica, Vol. 116, pp. 135 − 157, 1966.

[2]. Grue, J., D. Clamond, M. Huseby and A. Jensen: "Kinematics of Extreme Waves in Deep Water", Journal of Applied Ocean Research, Vol. 25, pp. 355 − 366, 2003

[4]. Gudmestad, O. T.:"A new Approach for Estimating Irregular Deep Water Wave Kinematics", Journal of Applied Ocean Research, Vol. 12, No. 1, pp. 19 − 24, 1990.

[5]. Gudmestad, O. T. and J. J. Connor: "Linearization Methods and the Influence of Current on the Nonlinear Hydrodynamic Drag Force", Journal of Applied Ocean Research, Vol. 5, No. 4, pp. 184 − 194, October 1983.

[6]. Gudmestad, O. T. and G. Moe: "Hydrodynamic Coefficients for Calculation of Hydrodynamic Loads on Offshore Truss Structures", Marine Structures, Vol. 9, No 8, pp. 745−758, Sept. 1996.

[7]. Gudmestad, O. T. and G.A. Poumbouras: "Time and Frequency Domain Wave Forces on Offshore Structures", Technical note in Journal of Applied Ocean Research, Vol. 10, No. 1, pp. 43 − 46, 1988.

[8]. Morison, J. R., M. P. O'Brian, J. W. Johnson, and S. A. Schaaf: "The forces exerted by surface waves on piles", petroleum Transactions, AIME, Vol. 189, pp.149 − 157, 1950.

[9]. Newman, J. N., O. Faltinsen and T. Vinje: "Nonlinear wave loads on a slender vertical cylinder," Journal of Fluid Mechanics, Vol. 289, pp. 179−198, 1995.

[10]. Sarpkaya, T. and M. St. Q. Issacsson: "Mechanics of Wave Forces on Offshore Structures", Van Nostrand Reinold, New York, 1981

[11]. Skjelbreia, J. E., G. Berek, Z. K. Bolen, O. T. Gudmestad, J. C. Heidemann, R. D. Ohmart, N. Spidsøe and A. Tørum: "Wave kinematics in Irregular Waves", Proceedings Vol. 1A of the 10th OMAE Conference, pp. 223 − 228, Stavanger, Norway

[12]. Stansberg, C. T. and O. T. Gudmestad: ”Nonlinear Random Wave kinematics Models Verified against Measurements in Steep Waves”, Proceedings Vol. 1A pp. 15 – 24 of 15th OMAE Conference, Florence, Italy, 1996

[13]. Stansberg, C. T.: "Comparing Ringing Loads from Experiments with Cylinders of Different Diameters – An Empirical Study", Proceedings (Vol. 2) of the 8th Int. Conference on the Behavior of Offshore Structures, BOSS’97, pp.95 – 109, Delft, the Netherlands, July 1997.

[14]. Wolfram, S. “The *Mathematica* 5”, Fifth Edition. Mathematica Version 5. Cambridge University Press. 2003. 1463 pp